

SECTION 6

PROGRAMMING

6-1. INTRODUCTION

Measurement and analysis (controlling the operation of the 4195A) can be performed with only a few key strokes by using the following capabilities of the HP 4195A.

- USER DEFINED FUNCTION
- SWEEP END FUNCTION
- USER PROGRAM (Auto Sequence Program: ASP)
- HP-IB function

This section describes these capabilities, their command syntax, and gives programming examples.

6-2. COMMAND SYNTAX

This paragraph describes the command syntax of the 4195A device dependent commands. The 4195A device dependent commands are unique commands used to control the 4195A's operation. Most of the commands which correspond to the keys and softkeys on the 4195A's front panel, are 4195A device dependent commands.

These commands are classified into the following six syntax types. The command syntax diagram is shown in Table 6-1.

1. Header only type
2. Numeric data type
3. Multiple numeric data type
4. String data type
5. Select type
6. Others

NOTE

The syntax type for any command can be found in APPENDIX E, COMMANDS LIST.

6-2-1. HEADER ONLY TYPE

Header Only Type Commands are constructed of a header only. The **AUTO**, and **COPY** commands are examples included in this type. The syntax diagram is shown in Type 1 of Table 6-1.

6-2-2. NUMERIC DATA TYPE

The Numeric Data Type Commands are constructed of a header, an equal sign (=), and a numeric expression. These commands are used to enter data into registers, as follows.

ex) **START= 100 MHZ**
STOP= MKR

The syntax diagram is shown in Type 2 of Table 6-1.

6-2-3. MULTIPLE NUMERIC DATA TYPE

The Multiple Numeric Data Type Commands are constructed of a header, an equal sign (=), and numerical expressions separated by commas. The **POINT**, and **PSCALE** commands, are included in this type. This type is classified into three types as shown in Type 3, 4 and 5 of Table 6-1.

6-2-4. STRING DATA TYPE

The String Data Type Commands are constructed of a header, and any characters enclosed by a pair of double or single quotation marks. This type includes the **CMT** and **PROG** commands, and is classified into three types as shown in Type 6, 7 and 8 of Table 6-1.

6-2-5. SELECT TYPE

The Select Type Commands are constructed of a header and a digit. This type includes the **FNC** and **ANA** commands, which are used as follows.

ex) **FNC2**
ANA1

The syntax diagram is shown in Type 9 of Table 6-1.

6-2-6. OTHERS

There are several commands which do not belong in any of the above syntax types, such as the **INPUT** and **OUTPUT** commands. The syntax diagrams for these commands are shown in Types 10 through 14 of Table 6-1.

Table 6-1. Command Syntax Diagram (1 of 2)

Syntax Type	Diagram
Type 1 (Header) ex) MKMX	
Type 2 (Numeric) ex) START	
Type 3 (Multiple) ex) OPNSTD	
Type 4 (Multiple) ex) POINT	
Type 5 (Multiple) ex) PSCALE	
Type 6 (String) ex) CMT	
Type 7 (String) ex) DISP	

Table 6-1. Command Syntax Diagram (2 of 2)

Syntax Type	Diagram
Type 8 (String) ex) PROG	
Type 9 (Select) ex) FNC	
Type 10 (Other) ex) INPUT	
Type 11 (Other) ex) OUTPUT	
Type 12 (Other) ex) EDIT	
Type 13 (Other) ex) DMA	
Type 14 (Other) ex) LMX	

6-3. USER DEFINED/SWEEP END FUNCTIONS

USER DEFINED FUNCTIONS and **SWEEP END FUNCTIONS** provide the ability to define single key stroke functions to replace multiple key and softkey combinations. The User Defined Function and the Sweep End Function are defined using the same procedure, but their operations are different; the User Defined Function works when the defined softkey is pressed, and the Sweep End Function is performed at the end of the sweep. These capabilities are used for the following uses.

- Press a single softkey to setup a frequently used measurement setup.
- Press a single softkey to read a parameter (ex. XdB Bandwidth, C/N, S/N, and etc.).
- Perform Alternate Sweep measurements, etc.

6-3-1. SOFTKEY/COMMAND FOR THE USER DEFINED/SWEEP END FUNCTION

The following paragraphs describe the softkeys/commands used for the User Defined Function, and the Sweep End Function. To display the softkey menus for these functions press the **USER DEFINE** key. Before reading the next paragraph, refer to the APPENDIX D, Softkey Tree.

6-3-2. PERFORM USER DEFINED FUNCTION/SWEEP END FUNCTION

Five User Defined Functions are available, and can be performed by using the ' 1 ' through ' 5 ' softkeys or by entering the **UDF1** through **UDF5** commands.

NOTE

A User Defined Function cannot be performed from within a User Program (ASP), a Sweep End Function, or from within User Defined Function, so the **UDF1** through **UDF5** commands cannot be used in a User Program (ASP), a Sweep End Function, or another User Defined Function.

The ' **A** ', ' **B** ' and ' **C** ' softkeys or the **SEFA1**, **SEFB1**, and **SEFC1** commands are used to perform the Sweep End Function. Three Sweep End Functions are available. The Sweep End Function can be performed at the end of every sweep, when the softkey label is green, or when a **SEFA1**, **SEFB1**, or **SEFC1** command is entered, will continue to be performed until a **SEFA0**, **SEFB0**, or **SEFC0** command is entered.

NOTE

When the **SEFA1** (or 0), **SEFB1** (or 0), and **SEFC1** (or 0) commands are used in a User Program, the function cannot be performed, but the on/off part of the softkey can be set. When a User Program is running, the Sweep End Function is not available.

6-3-3. DEFINE USER DEFINED FUNCTION/SWEEP END FUNCTION

The **'fctn1'** through **'fctn5'** softkeys, or the **DF1** through **DF5** commands are used to define a User Defined Function.

The **'fctnA'** through **'fctnC'** softkeys or the **DFA** through **DFC** commands are used to define the Sweep End Function.

When any of the these softkeys are pressed, the 4195A enters into the UDF editor mode, and the following commands are displayed on the keyboard input line.

```
DF1" " (DF2" ", DF3" ", DF4" ", or DF5" ")
or
DFA" " (DFB" ", or DFC" ")
```

The User Defined Function and the Sweep End Function are defined by entering the required commands for the function to be defined between double quotation marks (") and separated by semicolons (;). When the definition of the User Defined Function or Sweep End Function is completed, press the **ENTER/EXECUTE** key to store the function. The total number of characters entered between two double-quotation marks must be less than 83 characters. If the User Defined Function is defined using the "DF1" command, the ' 1 ' softkey will perform the defined function.

NOTE

Commands in which multi-statements are allowed, can be used to define User Defined Functions and Sweep End Functions. In APPENDIX E, Command List, commands in which multi-statements are not allowed, are marked with a bullet ' • '.

NOTE

All commands are entered using the alphabet keys, or by pressing the softkey or key which corresponds to the command. For example, the **MKMX** command can be entered by pressing the **'MKR→MAX'** softkey, or by entering **M, K, M,** and **X**. In the UDF editor mode, the function of the softkey is not performed but the corresponding command is displayed when the softkey is pressed.

6-3-4. EXIT USER DEFINED FUNCTION/SWEEP END FUNCTION EDITOR

When the definition of a User Defined Function or Sweep End Function is completed, the **ENTER/EXECUTE** key is pressed to enter the function. The 4195A will exit from the UDF editor mode when the **ENTER/EXECUTE** key is pressed.

If you want to exit from the UDF editor mode without storing the function, press the **'EXIT UDF edit'** softkey or the **'EXIT SEF edit'** softkey.

6-3-5. LABELING A DEFINED SOFTKEY

The softkey labels for the User Defined Functions and Sweep End Functions can be changed, the label length is 15 characters maximum.

To enter the softkey labels for the ' 1 ' through ' 5 ' softkeys, the 'fctn1 KEY LBL' through 'fctn5 KEY LBL' softkeys or the **LBL1** through **LBL5** commands are used.

To enter the softkey label of the ' A ' through ' C ' softkeys, the 'fctnA KEY LBL' through 'fctnC KEY LBL' softkeys or the **LBLA** through **LBLC** commands are used.

When any of these softkeys are pressed, the following corresponding messages are displayed on the keyboard input line.

LBL1" " (**LBL2" "**, **LBL3" "**, **LBL4" "**, or **LBL5" "**)
 or
LBLA" " (**LBLB" "** or **LBLC" "**)

The softkey is labeled by entering the characters, between two double-quotation (") marks.

NOTE

After the new label is entered, the letter or number character on the softkeys previous label (1, 2, 3, 4, 5, A, B, or C) remains on the new softkey label. For example, if you enter the **LBL1"XYZ"** command to label the UDF 1 softkey, the softkey label is changed from ' 1 ' to 'XYZ 1'.

NOTE

The function and label for User Defined Functions and Sweep End Functions are stored in the battery powered back-up memory. They can also be stored on a Flexible Disc, as a part of the instrument STATE.

6-3-6. Using Example

This paragraph describes how to use User Defined Functions, and Sweep End Functions.

The following three examples will be given.

Example 1: Measurement Condition Set-up (User Defined Function)

Example 2: -3 dB Bandwidth (User Defined Function)

Example 3: Signal Track Function (Sweep End Function)

Example 1: Define User Defined Function 1 (UDF 1) to set the measurement conditions, and then perform UDF 1

MEASUREMENT CONDITION:

MEASUREMENT FUNCTION:	SPECTRUM
MEASUREMENT UNIT:	dBm
TEST PORT:	T1
CENTER FREQUENCY:	100 MHz
SPAN FREQUENCY:	10 MHz
SOFTKEY LABEL:	SETTING NO.1

Table 6-2 lists the key strokes to define the function and label for example 1.

NOTE

To enter the alphabetical characters 'A' to 'U', the **blue** key must be pressed (blue key indicator on). To enter 'V' to 'Z', the **green** key must be pressed before the alphabetical key is pressed, for every keystroke.

Table 6-2. Procedure to Define the User Defined Function (Example 1)

Key Stroke	Display on Keyboard Input Line
<u>USER DEFINE</u> , 'DEFINE FCTN' , 'fctn 1'	DF1" "
<u>CONFIG</u> , 'SPECTRUM' , ; (or F,N,C,2 , ;)	DF1"FNC2;"
'PORT SELECT' , 'SPECTRUM' , 'T1' , ; (or P,O,R,T,2 , ;)	DF1"FNC2;PORT2;"
<u>FORMAT</u> , 'SPECTRUM' , 'dBm' , ; (or S,A,P,1 , ;)	DF1"FNC2;PORT2;SAP1;"
<u>CENTER</u> , 1,0,0,M , ; (or C,E,N,T,E,R,=,1,0,0,M , ;)	DF1"FNC2;PORT2;SAP1;CENTER=100M;"
<u>SPAN</u> , 1,0,M (or S,P,A,N,=,1,0,M)	DF1"FNC2;PORT2;SAP1;CENTER=100M; SPAN=10M"
<u>ENTER/EXECUTE</u>	(UDF1 is defined.)
<u>USER DEFINE</u> , 'KEY LBL entry' , 'fctn1 KEY LBL'	LBL1" "
S,E,T,T,I,N,G,(space),N,O,(period),1	LBL1"SETTING NO.1"
<u>ENTER/EXECUTE</u>	(Label '1' is changed to 'SETTING NO.1 1' .)
<u>USER DEFINE</u> , 'SETTING NO.1 1'	(UDF1 is performed.)

- NOTE:** 1. Under-lined characters indicate a key on the 4195A's front panel (ex: CONFIG indicates the **CONFIG** key).
2. Characters between two single-quotation marks indicate a softkey (ex: **'SPECTRUM'** indicates the **'SPECTRUM'** softkey).

Example 2: Define User Defined Function 2 (UDF 2) to get the -3 dB Bandwidth, and then perform UDF 2.

Table 6-3 shows the key strokes used to define the function and label for example 2.

Table 6-3. Procedure to Define the User Defined Function (Example 2)

Key Stroke	Display on Keyboard Input Line
<u>USER DEFINE</u> , 'DEFINE FCTN', 'fctn 2'	DF2" "
<u>MODE</u> , 'o&LCRS', ; (or M,C,F,4, ;)	DF2"MCF4;"
<u>MKR→</u> , 'Δ mode on off', 1, ; (or D,E,L,T,1, ;)	DF2"MCF4;DELT1;"
'WIDTH on off', 1, ; (or W,I,D,T,H,1, ;)	DF2"MCF4;DELT1;WIDTH1;"
'MKR→MAX', ; (or M,K,M,X, ;)	DF2"MCF4;DELT1;WIDTH1;MKMX;"
'Δ VALUE entry', - , 3 (or LCURS=-3)	DF2"MCF4;DELT1;WIDTH1;MKMX; LCURS=-3"
<u>ENTER/EXECUTE</u>	(UDF2 is defined.)
<u>USER DEFINE</u> , 'KEY LBL entry', 'fctn2 KEY LBL'	LBL2" "
-, 3, D, B, (space), B, A, N, D, W, I, D, T, H	LBL2"-3DB BANDWIDTH"
<u>ENTER/EXECUTE</u>	(Label '2' is changed to '- 3 DB BANDWIDTH 2'.)
<u>USER DEFINE</u> , '-3DB BANDWIDTH 2'	(UDF2 is performed.)

Example 3: Define a signal tracking function for Sweep End Function A (SEF A), and then perform SEF A

Table 6-4 shows the key strokes used to define the function and label for example 3.

Table 6-4. Procedure to Define the User Defined Function (Example 3)

Key Stroke	Display on Keyboard Input Line
<u>USER DEFINE</u> , 'SWP END FCTN', 'fctn A'	DFA" "
<u>MODE</u> , 'oMKR', ; (or M,C,F,1, ;)	DFA"MCF1;"
<u>MKR</u> →, 'MKR→MAX', ; (or M,K,M,X, ;)	DFA"MCF1;MKMX;"
'MKR→CENTER' (or M,K,C,T,R)	DFA"MCF1;MKMX;MKCTR"
<u>ENTER/EXECUTE</u>	(SEF A is defined.)
<u>USER DEFINE</u> , 'SWP END FCTN', 'LBL entry', 'fctnA KEY LBL'	LBLA" "
S,I,G,N,A,L,(space),T,R,A,C,K	LBLA"SIGNAL TRACK"
<u>ENTER/EXECUTE</u>	(Label 'A' is changed to 'SIGNAL TRACK A'.)
<u>USER DEFINE</u> , 'SIGNAL TRACK A'	(SEF A is activated.)

NOTE

Alternate sweep measurements are performed by defining and using two Sweep End Functions. The following example shows how to measure/display reflection/transmission characteristics. In this example, Sweep End Function A and Sweep End Function B are used.

Example:

DFA"RB=MA;DMB=RB;UNITB'DB';PRMB'T';MTHB1;PORT2;SEFA0;SEFB1"
DFB"RA=MA;DMA=RA;UNITA'DB';PRMA'R';MTHA1;PORT1;SEFB0;SEFA1"

LBLA"REFLECTION"
LBLB"TRANSMISSION"

6-4. USER PROGRAM (Auto Sequence Program: ASP)

A USER PROGRAM (ASP) is the 4195A's internal programming capability which makes it possible to automate the 4195A's operation, without an external controller.

6-4-1. COMMANDS USABLE IN ASP PROGRAMS

The commands used in User Programs are classified as follows.

1. 4195A Device Dependent Commands
2. 4195A BASIC Statements
3. 4195A Arithmetic Operators

1. 4195A Device Dependent Commands

Most of the 4195A's device dependent commands can be programmed. When in the User Program (ASP) editor mode, the softkey corresponding to commands are available, so a program can be edited by using the keystrokes used in normal operation.

NOTE

There are commands which cannot be used in a program. For example, the **UDF1** command cannot be used in a User Program. The details of each commands are described in Sections 4, 5, and in paragraph 6-2 through 6-3.

2. 4195A BASIC Statements

The following fifteen BASIC statements can be used in User Programs.

END, IF, THEN, FOR, TO, NEXT, GOTO, GOSUB, RETURN, AND, OR, PAUSE, BEEP, DISP, WAIT

The details of the 4195A's BASIC statements are described in paragraph 6-4-2, HP 4195A BASIC STATEMENTS.

3. 4195A Arithmetic Operators

All of the arithmetic operators listed in Table 5-1, Math Operators, can be used in User Programs. Refer to paragraph 5-3, MATH OPERATORS.

6-4-2. HP 4195A BASIC STATEMENTS


BASIC language statements which can be used in User Programs are introduced here. There are fifteen BASIC program statements, and corresponding softkeys (refer to APPENDIX D, Softkey Tree). These softkeys are displayed by pressing the **PROGRAM** key in the User Program (ASP) editor mode.

NOTE

To enter the ASP edit mode and display the softkey menu, press the **PROGRAM** key, 'EDIT' softkey and **ENTER/EXECUTE** key.

IF . . . THEN

This statement construct provides conditional branching.



ex) 10 IF A(100) > 5 THEN R0 = 1
 50 IF R0 = 1 AND R1 = 1 OR R2 = 1 THEN GOTO 50
 100 IF R0 AND R1 THEN R2 = 1


-- Boolean expression can include the following symbols, and statements.

= , <> , < , > , <= , >= , AND , OR

-- When **Rn** (n= 0 to 99) is entered as the boolean expression, and if **Rn** is equal to 0 (zero), the boolean expression is judged as false. If **Rn** is not equal to 0 (zero), the boolean expression is judged as true.

FOR . . . TO and NEXT

This construct defines a loop which is repeated until the loop counter passes a specific value.




```
ex)   10  FOR R5 = 1 TO 100 , 5
      :
      100 NEXT R5
```

- **Rn** (n= 0 to 99) should be used as the loop counter.
- When the step size is not defined, it is automatically set to either +1 or -1 according to the values input.
- Single variables **Rn** (n= 0 to 99) can be used as the initial value, final value, and step size. Single variables (MKR, START and etc.) can be used as the initial value.
- The maximum nesting level for the **FOR ... TO ... NEXT** construct is 10.

GOTO


GOTO transfers program execution to the specified line. The specified line must exist within the program.



```
ex)   100 GOTO 10      ( Jump to line number 10 )
```

GOSUB and RETURN

GOSUB and **RETURN** transfers program execution to a subroutine at the specified line. The specified line must exist within the program. The current program line is remembered in anticipation of a **RETURN** instruction.



```

ex)      :
          10 GOSUB 200      ( Jump to subroutine starting at line number 200 )
          20 SWTRIG
          :
          200 START = 1 MHZ ( Start of subroutine )
          :
          300 RETURN       ( End of subroutine. Return to line number 20 )
          :
  
```

-- The maximum nesting level for a **GOSUB ... RETURN** construct is 10.

AND and OR

AND is the logical-AND operator, and **OR** is the logical-OR operator. These statements can be used only within an **IF** construct.

PAUSE

PAUSE suspends program execution until one of the following program control commands are executed. Softkeys are provided for program control commands. The program control commands are explained in paragraph 6-4-3.

CONT(inue) Causes the program to continue at the next step.

RUN When the '**RUN**' softkey is pressed, the program will start over again from the beginning of the program.

STEP When the '**STEP**' softkey is pressed, the next program line will be executed.

STOP Press the '**STOP**' softkey to stop program execution.



```

ex)      50 PAUSE          ( Program execution is suspended here. )
  
```

NOTE

A User Program can be started by inputting a start signal through the PROGRAM START connector. Refer to paragraph 6-4-7, PROGRAM START CONNECTOR.

BEEP


BEEP causes the 4195A to emit an audible tone for 150 msec.



ex) 50 BEEP (The instrument will beep)

WAIT


WAIT causes the instrument to wait approximately the number of milliseconds specified by the number following the statement. The wait time range is from 0 to approximately 10 minutes. Setting resolution is 10 msec. If WAIT 153 is set, the wait time is rounded off to 150 msec.



ex) 50 WAIT 535 (wait time = 540 msec)

DISP

DISP will display either a comment or the contents of register **Rn** or both on the System Message line.




ex) 10 DISP"R1=" , R1 (Display 'R1= (contents of R1)')
 50 DISP" ABC " (Display 'ABC')

NOTE

DISP can be used not only in a User Program, but also all other modes. Refer to paragraph 5-16-2.

END

END marks the end of the program, and when it is executed the program will stop.



ex) 300 **END** (Program ends here)

! (Remark sign)

The remark sign (!) is used to input a comment on a program line.

ex) 100 **SWTRG ! SWEEP START**

6-4-3. PROGRAM CONTROL COMMANDS

The following nine commands are used to control User Programs (ASP), and are mainly used via HP-IB.

RUN, PSTOP, PPAUSE, CONT, PSTEP, EDIT, QUIT, SCRATCH, PROG

Except for the **PROG** and **QUIT** commands, these commands correspond to the softkeys displayed after the **PROGRAM** key is pressed, and they initiate the same action. **QUIT** corresponds to the **'QUIT editor'** softkey in the User Program (ASP) editor mode softkey menu. The **'EDIT'** and **'SCRATCH'** softkeys are not executed until the **ENTER/EXECUTE** key is pressed. Other softkey commands are executed immediately when pressed.

RUN ('RUN' softkey)

RUN starts execution of the program in the work area. A program will always start from the beginning. While the program is running, all softkeys and keys are deactivated, except for the **'STOP'** and **'PAUSE'** softkeys.

PSTOP ('STOP' softkey)

PSTOP terminates program execution. While in the **STOP** state, the **CONT** command is not effective, however the **STEP** command can be used to single step a program from the beginning of the program.

PPAUSE ('PAUSE' softkey)

PPAUSE suspends program execution. If the **CONT** or **STEP** commands are sent, program execution will start from the next line. If the **RUN** command is sent, the program will start from the beginning. All key and softkey inputs are accepted while in the **PAUSED** state.

CONT ('CONT' softkey)

CONT resumes execution of a paused program at the command after which the **PPAUSE** command was received. This command is effective only while in the **PAUSED** state.

PSTEP ('STEP' softkey)

PSTEP performs single step execution of a program. In the **STOP** state, **PSTEP** single steps the program from the beginning. In the **PAUSE** state, the **PSTEP** single steps a program starting at a specified line number.

EDIT ('EDIT' softkey)

EDIT is used to enter to the User Program (ASP) editor mode to edit a program. If you send the **EDIT** command, the cursor will appear at the top of the program on the edit page. If you send the **EDIT** command followed by a line number, the cursor will appear at the program line number specified.

NOTE

If the **EDIT** command is sent after the program is stopped by the **'STOP'** softkey or the **PSTOP** command, the cursor will be positioned on the program edit line which was being executed when the **PSTOP** command was sent.

QUIT ('QUIT editor' softkey)

QUIT is used to exit from the User Program (ASP) editor mode (quit the program edit).

SCRATCH ('SCRATCH' softkey)

SCRATCH is used to delete the program from the work area.

PROG

This command is used to enter a User Program (ASP) without using the editor. This command has the syntax shown in Type 8 of Table 6-1, paragraph 6-2. An ASP program can be edited by entering the following command, from the keyboard input line.

PROG " 100 SWTRG "

A User Program (ASP) can be entered via HP-IB. An example of entering a User Program (ASP) via HP-IB, is shown in Paragraph 6-5-8, Example 4.

6-4-4. HOW TO EDIT A USER PROGRAM (ASP)

1. How to Enter/Exit the Edit Mode

To enter to the User Program (ASP) editor mode (edit page), perform the following procedure.

1. Press the **PROGRAM** key, then the softkey menu to control the User Program (ASP) is displayed.
2. Press the **'EDIT'** softkey, "EDIT" will be displayed on the keyboard input line.
3. Press the **ENTER/EXECUTE** key. Then the 4195A's display is changed to the edit page, and the cursor appears at program line 10.

NOTE

If you want to edit line number 100 (or to start editing from line number 100), press the **'EDIT'** softkey, **1**, **0**, and **0**, in step 2. The edit page will be displayed with the cursor at program line 100, after performing step 3.

To exit from the User Program (ASP) editor mode, press the **'QUIT editor'** softkey.

2. How to Enter Commands

Most of the 4195A device dependent commands can be entered by pressing the softkey or key which corresponds to the command. All commands can be entered using the alphabet keys. An example of entering the User program (ASP) is shown in Table 6-5.

If a User Program (ASP) program is edited by using softkeys, the commands can be entered in the sequence same as would be used for normal front panel operation, but some of the softkey menus in the User Program (ASP) edit mode are different from the menus in normal operation.

For example, there is an additional softkey menu between the softkey menu which includes the **'PORT SELECT'** softkey and the menu used to select the test port (input terminal) in the User Program (ASP) editor mode. So, in line number 30 of Table 6-5, the **'SPECTRUM'** softkey has been pressed, after pressing the **'PORT SELECT'** softkey.

NOTE

To enter the characters **'A'** to **'U'**, the **blue** key must be turned on. To enter **'V'** to **'Z'**, the **green** key must be pressed before the alphabetical key is pressed.

Table 6-5. ASP Editing Example

PROGRAM	KEYSTROKE	
	SOFTKEY and KEY	ALPHABET KEY
10 FNC2	<u>CONFIG</u> , 'SPECTRUM', <u>ENT</u>	F, N, C, 2, <u>ENT</u>
20 RST	<u>PRESET</u> , <u>ENT</u>	R, S, T, <u>ENT</u>
30 PORT4	(<u>CONFIG</u>), 'PORT SELECT', 'SPECTRUM' 'T2', <u>ENT</u>	P, O, R, T, 4, <u>ENT</u>
40 CENTER=10MHZ	<u>CENTER</u> , 1, 0, <u>MHZ</u> , <u>ENT</u>	C, E, N, T, E, R, =, 1, 0, M, H, Z, <u>ENT</u>
50 SPAN=1MHZ	<u>SPAN</u> , 1, <u>MHZ</u> , <u>ENT</u>	S, P, A, N, =, 1, M, H, Z, <u>ENT</u>
60 SWTRG	<u>TRIG/RESET</u> , <u>ENT</u>	S, W, T, R, G, <u>ENT</u>
70 MKMX	<u>MKR</u> →, 'oMKR menu', 'MKR→MAX', <u>ENT</u>	M, K, M, X, <u>ENT</u>
80 END	<u>PROG</u> , 'END', <u>ENT</u>	E, N, D, <u>ENT</u>

- NOTE:**
1. ENT indicates the **ENTER/EXECUTE** key.
 2. Under-lined characters indicate a key on the 4195A's front panel.
 3. Characters enclosed in ' and ' (ex. 'END') indicate a softkey.

NOTE

When the 4195A is turned off, the User Program (ASP) in the work area is cleared. So, store the program on a flexible disc by using the internal flexible disc drive, or print the program listing. How to store the program is described in paragraph 5-18, and how to print the program listing is described in paragraph 5-13.

6-4-5. EDITING HINTS

This paragraph gives editing hints for User Program (ASP).

1. RST Command

RST resets the 4195A to the same state as would pressing the **PRESET** key. But, when **RST** is used in a User Program (ASP), the **SINGLE** sweep mode is set, whereas the **CONT**(inuous) sweep mode is set when the **PRESET** key is pressed while in the normal operation mode.

2. SWM1 Command

SWM1 sets the 4195A to the continuous sweep mode during normal operation. When the continuous sweep mode is used in a User Program (ASP), the sweep measurement will not be continuously performed. The sweep start timing is dependent on commands used latter in the program. So the recommend use for **SWM1** is to set the stimulus/receiver controls for manual (front panel) operation. When **SWM1** is executed within a program, the 4195A will start continuous sweep when the program is stopped or paused. Select the single sweep mode (**SWM2**) in an ASP program to make a sweep measurement.

3. SWTRG Command

SWTRG is used to start a sweep measurement. When **SWTRG** is used within a User Program, the 4195A's operation is different for the **TRGM1/TRGM2** trigger settings modes, as follows.

Internal trigger mode (**TRGM1**):

SWTRG starts a sweep measurement, and the next program line is executed after the sweep measurement is finished.

External trigger mode (**TRGM2**):

SWTRG arms the trigger to start a sweep, but does not start the sweep, and the next program line is executed. The sweep measurement is not performed. The measurement is started by **TRIG**, as shown in Figure 6-1.

Figure 6-1 shows program examples for a sweep measurement using these two trigger modes.

TRGM1	TRGM2
10 RST	10 RST
20 SWM2	20 SWM2
30 TRGM1	30 TRGM2
40 SWTRG	40 SWTRG
50 MKMX	50 FOR R0= 1 TO 401
60 END	60 TRIG
	70 NEXT R0
	80 MKMX
	90 END

Figure 6-1. Examples of the "SWTRG" Command

4. TRGM2 Command

TRGM2 sets the external trigger mode. When in a User Program , triggering by the '**PT MEAS TRIG**' softkey and from the **EXT TRIGGER** connector are not available. In a User Program, the '**CONT**' softkey and the **PROGRAM START** connector can be used instead, by using the following program.

```

:
100 PAUSE
110 TRIG
:

```

5. SEFA1, SEFB1, and SEFC1 Command

These commands are used to set the Sweep End Function to **on**, but the defined Sweep End Function is not executed even when the sweep measurement is completed. When the following example User Program (1) is executed, the Sweep End Function will not work. The program must be stopped to execute a Sweep End Function. Program example (2) shows how to perform the function of a Sweep End Function in an User Program.

:	:
:	:
:	:
100 SEFA1	100 SWM2
110 SWTRG	110 SWTRG
:	120 MKMX;MKCTR;
:	130 GOTO 110
:	:
:	:
(1)	(2)

In program (2), the same commands used to define the Sweep End Function should be entered at program line 120.

6. GET Command

GET is used within a User Program to load 'DATA' and 'PPT' from the 4195A's internal disc drive. But the 'STATE' and 'ASP' cannot be loaded using the **GET** command from within a program.

7. Entering a Select Type Command

When you press the softkey for a **select** type command, such as **DPA**, a message will be displayed on the System Message Line, such as:

on= 1 , off= 0

These messages are to remind you of what to enter with a command, in this case the digits **1** or **0** to be added to the command to select the states **on** and **off**.

8. Multi Statement

(1) The 4195A device dependent commands:

More than two commands can be programmed in one program line, this technique is called the multi-statement form. A semicolon (;) is the statement separator. The maximum length of a program statement line is 81 characters, including the separators, and spaces (except for the line number and a single space after the line number). But there are some commands which cannot be used in a multi-statement program line. Appendix E: COMMANDS LIST, lists the commands and identifies the commands which can be used in a multi-statement program line. An example of multi-statement input is shown next.

50 FNC1;SWM2;DPBO

(2) HP 4195A BASIC statements:

The **DISP**, and **BEEP** BASIC statements are permitted in multi-statement input. The **GOTO**, **GOSUB**, **RETURN**, **END** statements are permitted, if they are the last statement on a program line. In the **IF . . . THEN** construct, the statements after **THEN** can be multi-statement.

NOTE

Commands input in the multi-statement form will be automatically rearranged internally in the proper order, when the program line is executed.

9. Program Line Number

The usable program line numbers are 1 to 32767. The line number is automatically increased by 10 when entering a line. So, when you enter program line 10, line number 20 will be displayed next.

The maximum number of the characters that can be entered on a program line is 81, this includes the line number, separators, and spaces (except for a single space after the line number). The maximum number of the program lines available on an edit page is 300 lines.

10. Edit Mode

The following keys are used to edit programs.

CLR LINE, DEL CHAR, INS CHAR, RECALL, Arrow Keys (Right, Left, Up, Down)

CLR LINE is used to clear or erase a program line, except for the line number. To erase a program line, move the cursor to the program line to be erased, and press the **CLR LINE** key and then the **ENTER/EXECUTE** key.

The UP and DOWN arrow keys move the cursor up/down to change the program line being edited. After the green key is pressed, if the UP or DOWN key is pressed, the displayed program page is changed to the next or previous program page.

To insert a new program line between two existing program lines, use a line number which falls between the two existing program line numbers. For example to insert a new line between line numbers 10 and 20 enter a new program line number from 11 to 19.

To copy a program line, move the cursor to the line to be copied, change the line number to the line number you want to copy the line to, and press the **ENTER/EXECUTE** key.

NOTE

Saving the program on a flexible disc allows you to edit a User Program using the more powerful set of editing functions of an HP desk top computer. Refer to paragraph 5-18, MASS STORAGE.

NOTE

Program syntax errors are not detected when entering a program in the edit mode, program errors are detected at run time.

11. Printing Out Register Data, Measured Data, and Comments

To print the measurement results, comments, etc., directly to a printer, the 4195A provides query commands such as **DISP?**, **CMT?**, and the **SEND** command. Set the 4195A to TALK ONLY and the printer to LISTEN ONLY.

For example,

```

:
100 DISP " BANDWIDTH(HZ)= ", R1
110 DISP?
:
200 MKR?
:
300 SEND " ABC "
:

```

In the preceding program segment when line number 110 is executed, the contents specified by **DISP** will be printed out. When line number 200 is executed, the contents of the MKR register will be printed out. When line number 300 is executed, the characters between two double quotation marks will be printed out.

The details of the query commands are described in paragraph 6-5-3. For more information on the **SEND** command, refer to paragraph 6-5-8.

NOTE

When **DISP?** and **CMT?** are used, data is output in the ASCII format. When the (register)? command; **MKR?**, **R0?**, etc., are used, the data is output in the selected format using the **FMT1** (ASCII), **FMT2** (binary 64-bit), or **FMT3** (binary 32-bit) commands. Select ASCII format to print on a printer. The details of data output formats are described in paragraph 6-5-5.

12. Storing User Programs

The User Program on the work area is lost when the 4195A is turned off, so user programs should be saved on a flexible disc by using the 4195A's disc drive. For more detailed information refer to paragraph 5-18. MASS STORAGE.

6-4-6. HOW TO EXECUTE A USER PROGRAM (ASP)

This paragraph describes how to use an User Program (ASP).

1. Press the **PROGRAM** key to display the softkey menu used to control program execution.

NOTE

The command softkeys used to control program execution are described in paragraph 6-4-3, PROGRAM CONTROL COMMANDS.

2. Press the **'EDIT'** softkey and the **ENTER/EXECUTE** key, and confirm that the program you want to execute is in the work area.

NOTE

The program you want to execute must be in the work area. If the program is stored on a flexible disc, load it to the work area. Refer to paragraph 5-18. MASS STORAGE.

3. Press the **'QUIT editor'** softkey to exit from the editor mode.
4. Press the **'RUN'** softkey to start program execution.

NOTE

If **SEND**, **SENDPS**, **COPY**, or the query command are used in a User Program (ASP), set the 4195A to Talk-only, and connect the Listen-only device, printer, plotter, etc.

When Errors Occur:

Syntax is checked by the system interpreter during program execution, so syntax errors are caught at run time. In addition, setting errors, such as parameter range, and function mode are also checked at run time. When an error is found, the program stops, and a message is displayed on the System Message Line. The Error Messages are listed in Appendixes B, and C. The following is a typical error message.

Error Message	[Line Number]
ex) Multi statement not allowed	[100]

NOTE

If you press the **'EDIT'** softkey and the **ENTER/EXECUTE** key after the error is detected in a User Program (ASP) program, the cursor will be positioned at the program line where the error was detected.

6-4-7. PROGRAM START CONNECTOR

The Program Start Connector is used to input an external TTL level trigger pulse to start/continue User Programs (ASP).

A User Program (ASP) is triggered by the rising edge of a TTL level signal, or with a switch which is normally at ground and connects to 5 V through a pull-up resistor when activated to give the low-to-high transition (creates a rising edge). Refer to Figure 6-2 for the specifications of the input pulse.




Figure 6-2. Input Pulse

The following program is an example using the **PROGRAM START** connector.

```

10 DISP " PROGRAM STARTED " ; BEEP
20 WAIT 1000
30 DISP " PROGRAM PAUSED "
40 PAUSE
50 DISP " PROGRAM CONTINUED " ; BEEP
60 WAIT 1000
70 GOTO 30
80 END

```

In this example, the User Program is started at program line 10, by an input pulse from the **PROGRAM START** connector. After about 1 sec the program pauses at line 40, and continues from line 50 by the next input pulse. The program loops back to line 30 and repeats.

NOTE

The **EXT TRIG** connector cannot be used to input the trigger pulse during the execution of a User Program. The **PROGRAM START** connector must be used to input the external trigger pulse instead.

A program example for using an external trigger with a User Program is shown next.

```

10 TRGM2
20 R0=1
30 SWTRG
40 PAUSE
50 TRIG
60 IF R0=NOP THEN 20
70 R0=R0+1
80 GOTO 40
90 END

```

6-4-8. HOW TO MAKE A HARD COPY PROGRAM LISTING

A program listing can be printed out by using the procedure described in paragraph 5-13-5, Copy Procedure. This is supplementary information.

- (1) To display the program to be copied on the screen, enter the User Program (ASP) editor, by pressing the 'EDIT' softkey and the ENTER/EXECUTE key.

NOTE

If the program you want to copy is saved on a flexible disc, load the program into the work area. The program load procedure is described in paragraph 5-18, MASS STORAGE. Refer to paragraph 5-18.

- (2) Exit from the User Program (ASP) editor, by pressing the 'QUIT editor' softkey.
- (3) Select the PRINT mode and perform the Copy Procedure in paragraph 5-13-5.

6-4-9. SAMPLE PROGRAMS

Sample User Programs (ASP) are introduced in this section.

Example 1. Ripple Measurement (Network)

Example 2. C/N measurement (Spectrum)

Example 3. Set up the Programmed Points Table

Example 4. Define a User Defined Function, Sweep End Function, and User Math

EXAMPLE 1. This program is used to measure the ripple of a 100 MHz Band-Pass Filter (-3 dB Bandwidth= 500 kHz).

Program Listing:

```

10 CMT"RIPPLE MEAS."
20 FNC1
30 RST
40 GPP1;PORT1
50 CENTER=100 MHZ ;SPAN=500 KHZ
60 SWTRG
70 MCF2;LMX(A)
80 ARSTR;ANA1
90 IF SMKR=STOP THEN R1=0;GOTO 140
100 MKACTION;MKMX
110 MKACTION;MKMN
120 DELT1
130 R1=DMKRA
140 DISP "RIPPLE(DB)=",R1
150 END

```

Line No.	Description
10	Display "RIPPLE MEAS." in the Comment Area.
20	Select the NETWORK measurement function.
30	Initialize the 4195A.
40	Select the T/R(dB)- θ measurement using CHANNEL 1.
50	Set the sweep range using the center and span frequencies.
60	Make a single sweep measurement.
70	Move the o marker to left most peak position, and the * marker to the right most peak position.
80	Set the analysis range between two markers, and enable the partial analysis capability.
90	Display ' RIPPLE(DB)=0 ', if only a peak exists within the measurement range.
100	Activate the * marker, move the * marker to maximum point.
110	Activate the o marker, move the o marker to minimum point.
120	Set the Δ mode on.
130	Store the value of ripple in variable R1.
140	Display the value of ripple on the System Message Line.
150	END

EXAMPLE 2. This program measures the Carrier to Noise (C/N) ratio.

Program Listing:

```

10 CMT"C/N MEAS."
20 FNC2
30 RST
40 PORT2;SAP1
50 CENTER=10 MHZ ;SPAN=10 MHZ ;ATT1=50
60 VFTR1;RBW=30 KHZ
70 SWTRG
80 AUTO;TRGM2
90 R2=R1+1000000
100 MKMX;R1=MKR
110 FOR R0=1 TO 10
120   SWM3;MANUAL=R1
130   TRIG
140   MANUAL=R2
150   TRIG
160   MCF2;MKR=R1;SMKR=R2;DELT 1
170   R3=DMKRA/R0+R3*( 1-1/R0 )
180 NEXT R0
190 DISP "C/N(DBM)=",R3
200 END

```

Line No.	Description
20	Select the SPECTRUM measurement function
40	Set the unit to dBm, and select input port T1
50	Set the sweep range, and set the T1's attenuator to 50 dB
80	Perform auto scaling, and select the external trigger mode
90	Store 'R1+1000000' in variable R2
100	Move the o marker to the maximum point, store the o marker's sweep data in variable R1
110	Select the MANUAL sweep mode, and set the frequency point indicated by R1 to the measurement point
120	Measure the point specified by the "MANUAL=" command
130-140	Measure the point indicated by R2
160	Select the "o&* MKRS" mode, determine the C/N ratio using the maximum peak point and a point 1 MHz away from the maximum peak point
170	Calculate the average C/N ratio

EXAMPLE 3. This program sets the Programmed Points Sweep Table in steps of 80 kHz in the frequency range from 190 MHz to 198 MHz, and from 202 MHz to 210 MHz, and in steps of 20 kHz in the frequency range from 198 MHz to 202 MHz.

Program List:

```

10 FNC1
20 RST
30 !
40 PTN=1
50 PTCLR
60 PTSWP1
70 R1=190M
80 !
90 FOR R0=1 TO 401
100 POINT=R1
110 IF R0<101 OR R0>300 THEN R1=R1+80K;GOTO 130
120 R1=R1+20K
130 NEXT R0
140 !
150 BEEP
160 DISP"PPT SET COMPLETED"
170 END

```

Line No.	Description
10	Select the Network measurement function
40	Select Programmed Points Table 1.
50	Clear table 1.
60	Select frequency sweep.
100	Enter the sweep point indicated by R1.
110	If the point number is less than 101, or greater than 300, add 80000 to R1, and go to program line 130.
120	If the point number is between 101 and 300, add 20000 to R1.
130	Repeat the program lines 100 to 120, until the point number (R0) is 401.